
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of Rel-13 eMTC MAC test case 7.1.3.5a which is part of the LTE-A PRO test suite in the ‘iwd-TTCN3-B2016-06_D17wk10’ ATS delivery.

The test case can be demonstrated to run with two Rel-13 LTE-A PRO UEs (see section 5) on two independent Test Platforms. Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_7_1_3_5a
Test Group:
LTE_A_PRO\7_1\MAC_eMTC.ttcn
ATS Version:
iwd-TTCN3-B2016-06_D17wk10
System Simulator used:
R&S CMW500


+


Anite Conformance Toolset


UE used:
Qualcomm MDM 9207 /  MDM 9x07, Altair (a company of Sony Corp.) ALT 1210
Verification Status:
PASS


4. Corrections required for TC 7.1.3.5a
Change 1 – Correction to function  ‘f_TC_7_1_3_5a_EUTRA’
	Function name
	f_TC_7_1_3_5a_EUTRA()

	Reason for change
	1.   According to Note 1 mentioned in Table 7.1.3.5a.3.2-1 of TS 36.523-1, the SS should transmit the PDU in repetitions as per the DL_REPETITION_NUMBER provided by the lower layer in downlink DCI assignments. But in the current TTCN implementation there are no repetitions set.
2.   In the current TTCN implementation no CE mode specific default UL grant is assigned in the definition of specific RACH procedure. 
3.   According to the value of “pucch_NumRepetitionCE_Msg4_Level0_r13” signalled in SIB2-BR, the PUCCH ACK for Msg4 (RRC Connection Setup) will be sent as four PUCCH ACK REPETITIONs. However, in the current TTCN implementation this is not properly handled in step 20.
4.   At steps 9, 14 and 19 of the test procedure,the RRC Connection Setup message and contention resolution ID needs to be transmitted as a bundle of 4 repetitions according to Note 1 mentioned in Table 7.1.3.5a.3.2-1 of TS 36.523-1. However, in the current TTCN implementation this is considered.

	Summary of change
	1.   Defined new variable  v_RepetitionNumber and initialized it as ‘1’ which leads to PDSCH data repeated for 4 times and used the variable in RACH procedure and also in RRC Connection Setup PDSCH data.
2.   Defined new variable v_UplinkGrantCE_Mode and initialized the variable with function call f_EUTRA_SS_InitialiseUplinkGranteCE and also used the variable in the newly defined RACH procedure  cs_RachProcedureConfig_7135a   (… see change 2).
3.   Defined new variable v_PUCCH_Num_Repetition_Count and used it in step 20 to handle a UE sending all 4 PUCCH ACK Repetitions for Msg4.

R&S specific interpretation:  
When presenting the eMTC Test Model & ASP design, MCC160 explained in R5w170004 (slide 8) how repetitions can be handled and used for ASP ”RachPreamble_Type”. 
R&S believes that repetitions could similarily be addressed in other ASPs 
such as “L1Mac_IndicationControl_Type”. Currently when UL_HARQ is enabled, this ASP triggers SS to report all received UL_HARQ without taking repetitions into consideration. The TTCN of TC 7.1.3.5a – which deals with PUCCH repetition for Msg.4 – was therefore enhanced to handle such repetitions.
Note that the Test Purpose is not affected by the proposed enhancement.
4.   Configured the SS for transmitting the RRC Connection Setup message and contention resolution ID as a bundle of 4 repetitions according to Note 1 mentioned in Table 7.1.3.5a.3.2-1 of TS 36.523-1.

	TTCN module
	LTE_A_PRO\7_1\MAC_eMTC.ttcn

	MCC160 Comment
	1. Accepted
2. Accepted

3. Not Accepted. In accordance with the principle the UL repetitions at L1 has to be handled by SS and only one indication per bundle is forwarded to TTCN. ASP clarification is required, will be proposed in next work shop. MCC 160 agrees current SS reporting all repetitions does not impact TP.

4. Accepted

	
	


Before change
	function f_TC_7_1_3_5a_EUTRA() runs on EUTRA_PTC
  { 
    /* Correct HARQ process handling / CCCH/ Enhanced Coverage/ CE Mode A */
    var octetstring v_EncodedRRCConnectionSetup;
    var template (value) RachProcedureConfig_Type v_RachProcedureConfig;
    var template (omit) SRB_COMMON_IND v_SrbIndByRef:=omit;    
    // not used (needed when other information of the RRC message needs to be access)
    var template (omit) SYSTEM_IND v_SysIndByRef:=omit;        
    // to store SYSTEM_IND if it takes over the RRC message
    var boolean v_RRCConReq := false;
    var boolean v_HARQ_Ind := false;
    var boolean v_HARQ_Ind2 := false;
    var boolean v_RRCConSetupCmpl := false;
    var SRB_COMMON_IND v_SRB_COMMON_IND;
    var SYSTEM_IND v_SYSTEM_IND;
    var SubFrameTiming_Type v_Timing_SysInd;
    var SubFrameTiming_Type v_Timing_SysInd2;
    var SubFrameTiming_Type v_Timing_SrbIndReq;
    var SubFrameTiming_Type v_Timing_SrbIndCmpl;
    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var EUTRA_SecurityParams_Type v_Auth_Params; 
    timer t_Watchdog:=2.0; // General purpose wait timer set as T300 value
    ///////////////////////////////////////////////////////////////////////////////////
    //      PREAMBLE
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_Init(c1);
    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell1);
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    // Creates cell and performs registration
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);
    v_EncodedRRCConnectionSetup := 
    f_RRC_DL_CCCH_Encvalue(f_EUTRA_508_RRCConnectionSetup(eutra_Cell1));
    v_RachProcedureConfig := cs_RachProcedureConfig_7135(v_EncodedRRCConnectionSetup,
                             f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1), 
                             f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1));
                               /*cs_RachProcedureConfig_7135 has four entries
                                Step 3 CRC error
                                Step 9 not matching ContResolution ID
                                Step 14 CRC error
                                Step 19 success*/
    // Preconfigures SS for RACH response for RAR and contention resolution
    f_EUTRA_SS_ConfigRachProcedure(eutra_Cell1, omit, v_RachProcedureConfig);
    // Configure SS to enable report of PRACH preamble reception
    f_EUTRA_SS_ConfigRachPreambleIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);
    // to configure SS to start report reception of HARQ ACK or NACK
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);
    // Preconfig SS for default UL grant allocation on SR reception.
    f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(eutra_Cell1,cs_TimingInfo_Now, 
    cs_DciInfo_CcchDcchDtchUL_Def);
    // Test body starts
    f_EUTRA_TestBody_Set(true);
    ///////////////////////////////////////////////////////////////////////////////////
    //      Test Body
    //////////////////////////////////////////////////////////////////////////////////
    v_Auth_Params := f_EUTRA_Security_Get();
    //@siclog "Step 1" siclog@
    f_EUTRA_UE_Page_Def(eutra_Cell1);
    //@siclog "Step 2" siclog@
    t_Watchdog.start;
    SYSIND.receive(car_PRACH_Preamble_IND_Any(eutra_Cell1, ?));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");
    t_Watchdog.stop;
    //@siclog "Step 3" siclog@
    // Automatically taken care of
    //@siclog "Steps 4 - 5" siclog@
    t_Watchdog.start;
    alt {
      []SYSIND.receive(car_UL_HARQ_IND (eutra_Cell1, ?, ?))
        {
          t_Watchdog.stop;
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4: UE sending a ACK/NACK");
        }
      []SYSIND.receive(car_PRACH_Preamble_IND_Any (eutra_Cell1,?))
        {
          t_Watchdog.stop;
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");
        }
    }
    //@siclog "Step 6" siclog@
    // Automatically taken care of
    //@siclog "Steps 7 - 10" siclog@
    // step 7 and step 10 to match in default handler.
    //@siclog "Step 8" siclog@
    f_EUTRA_RacingCond_AwaitRrcMessageBeforeSysInd(car_SRB0_RrcPdu_IND(eutra_Cell1, 
                   cr_RRCConnectionRequest(?)),v_SrbIndByRef,v_SysIndByRef);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");
    //@siclog "Step 9" siclog@
    // Automatically taken care off
    // If step 11 not received then receive else to match
    f_EUTRA_RacingCond_MatchSysInd(v_SysIndByRef, car_PRACH_Preamble_IND_Any (eutra_Cell1, ?));
    //@siclog "Step 12" siclog@
    // Automatically taken care of
    // Note Racing should not be between steps 13 and 16 as UE has to wait for mac-
    ContentionResolutionTimer  which is 48 ms
    // before initiating the next PRACH procedure.
    //@siclog "Step 13" siclog@
    t_Watchdog.start;
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 13");
    t_Watchdog.stop;
    //@siclog "Step 14" siclog@
    // Automatically taken care of
    //@siclog "Steps 15 - 16" siclog@
    t_Watchdog.start;
    alt {
      []SYSIND.receive(car_UL_HARQ_IND(eutra_Cell1, ?, nack))
        {
          t_Watchdog.stop;
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 15: UE sending a NACK");
        }
      []SYSIND.receive(car_PRACH_Preamble_IND_Any (eutra_Cell1,?))
        {
          t_Watchdog.stop;
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 16");
        }
    }
    //@siclog "Step 17" siclog@
    // Automatically taken care of
    t_Watchdog.start;
    alt {
      //@siclog "Step 18" siclog@
      //@siclog "Step 19" siclog@
      // Automatically taken care of
      [not v_RRCConReq] SRB.receive(car_SRB0_RrcPdu_IND(eutra_Cell1, cr_RRCConnectionRequest(?))) 
      -> value v_SRB_COMMON_IND
        {
          v_Timing_SrbIndReq := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame; 
          // timestamp when RRCConnectReq has been received
          v_RRCConReq := true;
          if (not (v_HARQ_Ind and v_RRCConSetupCmpl and v_HARQ_Ind2)) {
            repeat;
          }
        }
      //@siclog "Step 20" siclog@
      [not v_HARQ_Ind] SYSIND.receive(car_UL_HARQ_IND (eutra_Cell1, ?, ack)) -> value v_SYSTEM_IND
        {
          v_HARQ_Ind := true;
          v_Timing_SysInd := v_SYSTEM_IND.Common.TimingInfo.SubFrame;     
          // timestamp when SYSTEM_IND has been received 
          if (not (v_RRCConReq and v_RRCConSetupCmpl and v_HARQ_Ind2) ) {
            repeat;
          }
        }
      //@siclog "Step 21" siclog@
      [not v_RRCConSetupCmpl] SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_Cell1, 
      cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),    
      cr_NAS_Indication(tsc_SHT_IntegrityProtected, 

      cr_508_SERVICE_REQUEST(v_Auth_Params.KSIasme))))-> value v_SRB_COMMON_IND
        {
          v_Timing_SrbIndCmpl := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame; 
          // timestamp when RRCConnectSetup complete has been received
          v_RRCConSetupCmpl := true;
          v_NAS_Ind := v_SRB_COMMON_IND.Signalling.Nas[0];
          if (not(v_RRCConReq and v_HARQ_Ind and v_HARQ_Ind2)) {
          // RRC connection shall be received before RRC connection setup complet;
            repeat;
          }
          // checked to guarantee run time errors if any un initialised timing info check 
          in if statement below;
        }
      [not v_HARQ_Ind2] SYSIND.receive(car_UL_HARQ_IND (eutra_Cell1, ?, ack)) -> value v_SYSTEM_IND
        // HARQ for RLC Status PDU for RRC Connection Setup Complete
        {
          v_HARQ_Ind2 := true;
          v_Timing_SysInd2 := v_SYSTEM_IND.Common.TimingInfo.SubFrame;     
           // timestamp when SYSTEM_IND has been received
          if (not(v_RRCConReq and v_HARQ_Ind and v_RRCConSetupCmpl)) {
          // note HARQ for Setup shall come before HARQ for HARQ for Status PDU
            repeat;
          }
        }
    }
    t_Watchdog.stop;
    if (not f_EUTRA_RacingCond_CheckSequence({v_Timing_SrbIndReq, v_Timing_SysInd, 
    v_Timing_SrbIndCmpl, v_Timing_SysInd2})) {
      f_SetVerdictInconc(__FILE__, __LINE__, "Step 18 - 20 - 21: got unexpected message order");
    }
    else {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 18 - 20 -21 ");
    }
    // to configure SS to stop report reception of HARQ ACK or NACK
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1,tsc_L1Mac_IndicationMode_Disable);
    //@siclog "Steps 22 - 23 - 24 - 25" siclog@
    // Steps 6 to 9 of the generic radio bearer establishment procedure (TS 36.508 5.4.3.3-1) are 
    Executed
    // to successfully complete the service request procedure
    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);
    f_EUTRA_ServiceReqPostamble_508Steps6_9 (eutra_Cell1, v_NAS_Ind);
    //test body ends
    f_EUTRA_TestBody_Set(false);
    ///////////////////////////////////////////////////////////////////////////////////
    //      Post Amble
    //////////////////////////////////////////////////////////////////////////////////
    // Configure SS to disable report of  PRACH preamble reception
    f_EUTRA_SS_ConfigRachPreambleIndMode (eutra_Cell1 ,tsc_L1Mac_IndicationMode_Disable);
    // Preconfigures SS for RACH response for default RAR and contention resolution
    f_EUTRA_SS_ConfigRachProcedure_Def(eutra_Cell1);
    //Switch off UE and expect a DETACH REQUEST message
    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);
  }



After  change
	function f_TC_7_1_3_5a_EUTRA() runs on EUTRA_PTC
  { 
    /* Correct HARQ process handling / CCCH/ Enhanced Coverage/ CE Mode A */
    var octetstring v_EncodedRRCConnectionSetup;
    var template (value) RachProcedureConfig_Type v_RachProcedureConfig;
    var template (omit) SRB_COMMON_IND v_SrbIndByRef:=omit;    
    // not used (needed when other information of the RRC message needs to be access)
    var template (omit) SYSTEM_IND v_SysIndByRef:=omit;        
    // to store SYSTEM_IND if it takes over the RRC message
    var boolean v_RRCConReq := false;
    var boolean v_HARQ_Ind := false;
    var boolean v_HARQ_Ind2 := false;
    var boolean v_RRCConSetupCmpl := false;
    var SRB_COMMON_IND v_SRB_COMMON_IND;
    var SYSTEM_IND v_SYSTEM_IND;
    var SubFrameTiming_Type v_Timing_SysInd;
    var SubFrameTiming_Type v_Timing_SysInd2;
    var SubFrameTiming_Type v_Timing_SrbIndReq;
    var SubFrameTiming_Type v_Timing_SrbIndCmpl;
    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var EUTRA_SecurityParams_Type v_Auth_Params; 
    var  RepetitionNumber_Type v_RepetitionNumber :=1;
    var integer  v_PUCCH_Num_Repetition_Count := 0; 
    var template (omit) UplinkGrantCE_Mode_Type   v_UplinkGrantCE_Mode;
    timer t_Watchdog:=2.0; // General purpose wait timer set as T300 value
    ///////////////////////////////////////////////////////////////////////////////////
    //      PREAMBLE
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_Init(c1);
    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell1);
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    // Creates cell and performs registration
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);
    v_EncodedRRCConnectionSetup := 
    f_RRC_DL_CCCH_Encvalue(f_EUTRA_508_RRCConnectionSetup(eutra_Cell1));
    v_UplinkGrantCE_Mode := f_EUTRA_SS_InitialiseUplinkGranteCE(eutra_Cell1); 
    v_RachProcedureConfig := cs_RachProcedureConfig_7135a(v_EncodedRRCConnectionSetup,
                             f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1), 
                             f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1) , 
                             v_UplinkGrantCE_Mode,v_RepetitionNumber);
                               /*cs_RachProcedureConfig_7135 has four entries
                                Step 3 CRC error
                                Step 9 not matching ContResolution ID
                                Step 14 CRC error
                                Step 19 success*/
    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1,
    cs_TimingInfo_Now,cs_DciDLInfo_Auto(fl_DciDlInfo_Init(e_DlSchedulingClass_UeSpecific,

    f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1),-,v_RepetitionNumber))));
    // Preconfigures SS for RACH response for RAR and contention resolution
    f_EUTRA_SS_ConfigRachProcedure(eutra_Cell1, omit, v_RachProcedureConfig);
    // Configure SS to enable report of PRACH preamble reception
    f_EUTRA_SS_ConfigRachPreambleIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);
    // to configure SS to start report reception of HARQ ACK or NACK
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);
    // Preconfig SS for default UL grant allocation on SR reception.
   f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(eutra_Cell1,cs_TimingInfo_Now, 
   cs_DciInfo_CcchDcchDtchUL_Def);
    // Test body starts
    f_EUTRA_TestBody_Set(true);
    ///////////////////////////////////////////////////////////////////////////////////
    //      Test Body
    //////////////////////////////////////////////////////////////////////////////////
    v_Auth_Params := f_EUTRA_Security_Get();
    //@siclog "Step 1" siclog@
    f_EUTRA_UE_Page_Def(eutra_Cell1);
    //@siclog "Step 2" siclog@
    t_Watchdog.start;
    SYSIND.receive(car_PRACH_Preamble_IND_Any(eutra_Cell1, ?));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");
    t_Watchdog.stop;
    //@siclog "Step 3" siclog@
    // Automatically taken care of
    //@siclog "Steps 4 - 5" siclog@
    t_Watchdog.start;
    alt {
      []SYSIND.receive(car_UL_HARQ_IND (eutra_Cell1, ?, ?))
        {
          t_Watchdog.stop;
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4: UE sending a ACK/NACK");
        }
      []SYSIND.receive(car_PRACH_Preamble_IND_Any (eutra_Cell1,?))
        {
          t_Watchdog.stop;
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");
        }
    }
    //@siclog "Step 6" siclog@
    // Automatically taken care of
    //@siclog "Steps 7 - 10" siclog@
    // step 7 and step 10 to match in default handler.
    //@siclog "Step 8" siclog@
    f_EUTRA_RacingCond_AwaitRrcMessageBeforeSysInd(car_SRB0_RrcPdu_IND(eutra_Cell1, 
                   cr_RRCConnectionRequest(?)),v_SrbIndByRef,v_SysIndByRef);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");
    //@siclog "Step 9" siclog@
    // Automatically taken care off
    // If step 11 not received then receive else to match
    f_EUTRA_RacingCond_MatchSysInd(v_SysIndByRef, car_PRACH_Preamble_IND_Any (eutra_Cell1, ?));
    //@siclog "Step 12" siclog@
    // Automatically taken care of
    // Note Racing should not be between steps 13 and 16 as UE has to wait for mac-
    ContentionResolutionTimer  which is 48 ms
    // before initiating the next PRACH procedure.
    //@siclog "Step 13" siclog@
    t_Watchdog.start;
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 13");
    t_Watchdog.stop;
    //@siclog "Step 14" siclog@
    // Automatically taken care of
    //@siclog "Steps 15 - 16" siclog@
    t_Watchdog.start;
    alt {
      []SYSIND.receive(car_UL_HARQ_IND(eutra_Cell1, ?, nack))
        {
          t_Watchdog.stop;
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 15: UE sending a NACK");
        }
      []SYSIND.receive(car_PRACH_Preamble_IND_Any (eutra_Cell1,?))
        {
          t_Watchdog.stop;
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 16");
        }
    }
    //@siclog "Step 17" siclog@
    // Automatically taken care of
    t_Watchdog.start;
    alt {
      //@siclog "Step 18" siclog@
      //@siclog "Step 19" siclog@
      // Automatically taken care of
      [not v_RRCConReq] SRB.receive(car_SRB0_RrcPdu_IND(eutra_Cell1, cr_RRCConnectionRequest(?))) 
      -> value v_SRB_COMMON_IND
        {
          v_Timing_SrbIndReq := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame; 
          // timestamp when RRCConnectReq has been received
          v_RRCConReq := true;
          if (not (v_HARQ_Ind and v_RRCConSetupCmpl and v_HARQ_Ind2)) {
            repeat;
          }
        }
      //@siclog "Step 20" siclog@
      [not v_HARQ_Ind] SYSIND.receive(car_UL_HARQ_IND (eutra_Cell1, ?, ack)) -> value v_SYSTEM_IND
        {
          v_HARQ_Ind := true;
          v_Timing_SysInd := v_SYSTEM_IND.Common.TimingInfo.SubFrame;     
          // timestamp when SYSTEM_IND has been received 
          if (not (v_RRCConReq and v_RRCConSetupCmpl and v_HARQ_Ind2) ) {
            v_PUCCH_Num_Repetition_Count:= v_PUCCH_Num_Repetition_Count+1;
            repeat;
          }
        }
      [not v_RRCConSetupCmpl and (v_PUCCH_Num_Repetition_Count < 4)] SYSIND.receive(car_UL_HARQ_IND (eutra_Cell1, ?, ack)) -> value v_SYSTEM_IND

         {

            v_PUCCH_Num_Repetition_Count:= v_PUCCH_Num_Repetition_Count+1;

            repeat;         

         }
      //@siclog "Step 21" siclog@
      [not v_RRCConSetupCmpl] SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_Cell1, 
      cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),    
      cr_NAS_Indication(tsc_SHT_IntegrityProtected, 

      cr_508_SERVICE_REQUEST(v_Auth_Params.KSIasme))))-> value v_SRB_COMMON_IND
        {
          v_Timing_SrbIndCmpl := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame; 
          // timestamp when RRCConnectSetup complete has been received
          v_RRCConSetupCmpl := true;
          v_NAS_Ind := v_SRB_COMMON_IND.Signalling.Nas[0];
          if (not(v_RRCConReq and v_HARQ_Ind and v_HARQ_Ind2)) {
          // RRC connection shall be received before RRC connection setup complet;
            repeat;
          }
          // checked to guarantee run time errors if any un initialised timing info check 
          in if statement below;
        }
      [not v_HARQ_Ind2] SYSIND.receive(car_UL_HARQ_IND (eutra_Cell1, ?, ack)) -> value v_SYSTEM_IND
        // HARQ for RLC Status PDU for RRC Connection Setup Complete
        {
          v_HARQ_Ind2 := true;
          v_Timing_SysInd2 := v_SYSTEM_IND.Common.TimingInfo.SubFrame;     
           // timestamp when SYSTEM_IND has been received
          if (not(v_RRCConReq and v_HARQ_Ind and v_RRCConSetupCmpl)) {
          // note HARQ for Setup shall come before HARQ for HARQ for Status PDU
            repeat;
          }
        }
    }
    t_Watchdog.stop;
    if (not f_EUTRA_RacingCond_CheckSequence({v_Timing_SrbIndReq, v_Timing_SysInd, 
    v_Timing_SrbIndCmpl, v_Timing_SysInd2})) {
      f_SetVerdictInconc(__FILE__, __LINE__, "Step 18 - 20 - 21: got unexpected message order");
    }
    else {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 18 - 20 -21 ");
    }
    // to configure SS to stop report reception of HARQ ACK or NACK
    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1,tsc_L1Mac_IndicationMode_Disable);
    //@siclog "Steps 22 - 23 - 24 - 25" siclog@
    // Steps 6 to 9 of the generic radio bearer establishment procedure (TS 36.508 5.4.3.3-1) are 
    executed
    // to successfully complete the service request procedure
    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);
    f_EUTRA_ServiceReqPostamble_508Steps6_9 (eutra_Cell1, v_NAS_Ind);
    //test body ends
    f_EUTRA_TestBody_Set(false);
    ///////////////////////////////////////////////////////////////////////////////////
    //      Post Amble
    //////////////////////////////////////////////////////////////////////////////////
    // Configure SS to disable report of  PRACH preamble reception
    f_EUTRA_SS_ConfigRachPreambleIndMode (eutra_Cell1 ,tsc_L1Mac_IndicationMode_Disable);
    // Preconfigures SS for RACH response for default RAR and contention resolution
    f_EUTRA_SS_ConfigRachProcedure_Def(eutra_Cell1);
    //Switch off UE and expect a DETACH REQUEST message
    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);
  }



Change 2 – Correction to template  ‘cs_RachProcedureConfig_7135a’
	Template name
	cs_RachProcedureConfig_7135a ()

	Reason for change
	In this specifc test scenario, RAR needs to be repeated and this needs to be indicated in the downlink assignmnent field of “REPETITION NUMBER”. 
Also the uplink grant assigned are to be CE mode specifc and the existing defined template cs_RachProcedureConfig_7135 can’t be used.

	Summary of change
	Defined the new template cs_RachProcedureConfig_7135a with the enhancement of paramters to assign the values for CE Mode specific uplink grants and  repetition number.

	TTCN module
	Common\EUTRA_Templates\EUTRA_CellCfg_Templates.ttcn

	MCC160 Comment
	Accepted in principle and enhanced cs_RachProcedureConfig_7135 instead of adding the new template cs_RachProcedureConfig_7135a


Before change
	


After change
	template (value) RachProcedureConfig_Type cs_RachProcedureConfig_7135a(octetstring p_RLC_PDU,
                                          RACH_TimingAdvance_Type p_RAR_TA,
                                          Dl_Bandwidth_Type p_Dl_Bandwidth, 

                                          template (omit) UplinkGrantCE_Mode_Type  

                                          p_UplinkGrantCE_Mode:=omit,

                                          template (omit) RepetitionNumber_Type 

                                          p_RepetitionNumber:=fl_RepetitionNumber_Init() ) := 
  { 

    /* @status    APPROVED (LTE) */
    RACH_ConfigCommon := omit, // not change feom initial value
    RACH_ConfigDedicated := omit,
    RachProcedureList := {
      {
        RAResponse := cs_RandomAccessResponseConfig_CRC_Error(p_Dl_Bandwidth,p_RepetitionNumber), 

        // step 3 CRC error// @sic R5-103846 sic@
        ContentionResolutionCtrl := cs_ContReslCtrl_None_TCRNTI
      },
      { 
        RAResponse := cs_RandomAccessResponseConfig_Def (p_RAR_TA,p_Dl_Bandwidth,-,-,

                                                         p_UplinkGrantCE_Mode,p_RepetitionNumber),
        ContentionResolutionCtrl := cs_ContentionResolutionCtrl_ContRLCPDU_Fail(p_RLC_PDU) 

        // Step 9 not matching ContResolution ID
      },
      {
       RAResponse := cs_RandomAccessResponseConfig_Def (p_RAR_TA,p_Dl_Bandwidth,-,-,

                                                        p_UplinkGrantCE_Mode,p_RepetitionNumber),
        ContentionResolutionCtrl := cs_ContentionResolutionCtrl_ContRLCPDU_CRC_Error(p_RLC_PDU) 

        // Step 19 CRC error
      },
      {
        RAResponse := cs_RandomAccessResponseConfig_Def (p_RAR_TA,p_Dl_Bandwidth,-,-,

                                                         p_UplinkGrantCE_Mode,p_RepetitionNumber),
        ContentionResolutionCtrl := cs_ContentionResolutionCtrl_ContRLCPDU_Def(p_RLC_PDU) 

       // Step 19 success
      }
    }
  };




Change 3 – Correction to template  ‘cs_RandomAccessResponseConfig_Def’
	Template name
	cs_RandomAccessResponseConfig_Def ()

	Reason for change
	In this test scenario, RAR needs to be repeated and this needs to be indicated in the downlink assignmnent field of  “REPETITION NUMBER”. 

In the current TTCN implementation, there is no option to provide explicit  Repetition number in template cs_RandomAccessResponseConfig_Def.

	Summary of change
	Enhanced the template cs_RandomAccessResponseConfig_Def with the parameter RepetitionNumber as a template argument and assigned the received  value to RepetitionNumber argument to template cs_DciInfo_RandomAccessResponse.
See also change 5.

	TTCN module
	Common\EUTRA_Templates\EUTRA_CellCfg_Templates.ttcn

	MCC160 Comment
	Accepted


Before change
	template (value) RandomAccessResponseConfig_Type          

                 cs_RandomAccessResponseConfig_Def(RACH_TimingAdvance_Type p_RAR_TA, 
                 Dl_Bandwidth_Type p_Dl_Bandwidth,integer p_RA_GrantRIV := 0,
                 integer p_RA_GrantIMCS := 15,
                 template (omit) UplinkGrantCE_Mode_Type   p_UplinkGrantCE_Mode := omit
  { /* ASP for initial configuration of a cell */
    /* @sic R5-103846: adding Dl_Bandwidth and sysinfo combination parameters sic@ */
    /* @sic R5s100430 Additional Changes: merge of cs_RandomAccessResponseConfig_InitialAccess_Def 
       and cs_RandomAccessResponseConfig_InitialAccess_Grant sic@ */
    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, 
        LTE_IRAT, POS) */
    Ctrl := {
      DciInfo := cs_DciInfo_RandomAccessResponse(p_Dl_Bandwidth),
      Rar := {
        List := {
          {
            RapId := {
              Automatic := true
            },
            InitialGrant := cs_InitialGrant_Def(p_RA_GrantRIV, p_RA_GrantIMCS),   
           // @sic R5s100430 Additional Changes sic@
            TimingAdvance := p_RAR_TA,
            TempC_RNTI := cs_TempC_RNTI_SameAsC_RNTI,
            InitialGrantCE_Mode := p_UplinkGrantCE_Mode // @sic R5w160205 R5-167569 sic@
          }
        }
      },
      BackoffInd := cs_BI_None
    }
  };



After change
	template (value) RandomAccessResponseConfig_Type          

                 cs_RandomAccessResponseConfig_Def(RACH_TimingAdvance_Type p_RAR_TA, 
                 Dl_Bandwidth_Type p_Dl_Bandwidth,integer p_RA_GrantRIV := 0,
                 integer p_RA_GrantIMCS := 15,
                 template (omit) UplinkGrantCE_Mode_Type   p_UplinkGrantCE_Mode := omit,

                 template (omit) RepetitionNumber_Type p_RepetitionNumber :=    

                                                       fl_RepetitionNumber_Init()) :=
  { /* ASP for initial configuration of a cell */
    /* @sic R5-103846: adding Dl_Bandwidth and sysinfo combination parameters sic@ */
    /* @sic R5s100430 Additional Changes: merge of cs_RandomAccessResponseConfig_InitialAccess_Def 
       and cs_RandomAccessResponseConfig_InitialAccess_Grant sic@ */
    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, 
        LTE_IRAT, POS) */
    Ctrl := {
      DciInfo := cs_DciInfo_RandomAccessResponse(p_Dl_Bandwidth,p_RepetitionNumber),
      Rar := {
        List := {
          {
            RapId := {
              Automatic := true
            },
            InitialGrant := cs_InitialGrant_Def(p_RA_GrantRIV, p_RA_GrantIMCS),   
           // @sic R5s100430 Additional Changes sic@
            TimingAdvance := p_RAR_TA,
            TempC_RNTI := cs_TempC_RNTI_SameAsC_RNTI,
            InitialGrantCE_Mode := p_UplinkGrantCE_Mode // @sic R5w160205 R5-167569 sic@
          }
        }
      },
      BackoffInd := cs_BI_None
    }
  };



Change 4 – Correction to template  ‘cs_RandomAccessResponseConfig_CRC_Error’
	Template name
	cs_RandomAccessResponseConfig_CRC_Error ()

	Reason for change
	In this test scenario, RAR needs to be repeated and this needs to be indicated in the downlink assignmnent field of  “REPETITION NUMBER”. 

In the current TTCN implementation, there is no option to provide explicit  repetition number and the default value assigned for repetition number in the template cs_RandomAccessResponseConfig_CRC_Error is always zero.

	Summary of change
	Enhanced the template cs_RandomAccessResponseConfig_CRC_Error with parameter RepetitionNumber as a template argument and assigned the received  value to template cs_DciInfo_RandomAccessResponse.
See also change 5.

	TTCN module
	Common\EUTRA_L2\EUTRA_MAC_Common.ttcn

	MCC160 Comment
	Accepted


Before change
	template (value) RandomAccessResponseConfig_Type cs_RandomAccessResponseConfig_CRC_Error( Dl_Bandwidth_Type p_Dl_Bandwidth:=  { // will cause CRC error on UE side
    /* @status    APPROVED (LTE) */
    Ctrl_CRC_Error  := {
      DciInfo := cs_DciInfo_RandomAccessResponse( p_Dl_Bandwidth),
      Rar := {
        List := {
          cs_RandomAccessResponseParameters_Def(cs_RapIdCtrl_Auto, cs_TempC_RNTI_SameAsC_RNTI)
        }
      },
      BackoffInd := cs_BI_None
    }
  };



After change
	template (value) RandomAccessResponseConfig_Type cs_RandomAccessResponseConfig_CRC_Error( Dl_Bandwidth_Type p_Dl_Bandwidth,template (omit) RepetitionNumber_Type p_RepetitionNumber := fl_RepetitionNumber_Init()) :=
  { // will cause CRC error on UE side
    /* @status    APPROVED (LTE) */
    Ctrl_CRC_Error  := {
      DciInfo := cs_DciInfo_RandomAccessResponse( p_Dl_Bandwidth,p_RepetitionNumber),
      Rar := {
        List := {
          cs_RandomAccessResponseParameters_Def(cs_RapIdCtrl_Auto, cs_TempC_RNTI_SameAsC_RNTI)
        }
      },
      BackoffInd := cs_BI_None
    }
  };



Change 5 – Correction to template  ‘cs_DciInfo_RandomAccessResponse’
	Template name
	cs_DciInfo_RandomAccessResponse ()

	Reason for change
	In this test scenario, RAR needs to be repeated and this needs to be indicated in the downlink assignmnent field of “REPETITION NUMBER”. 

However, in the current TTCN implementation, there is no option to provide explicit repetition number and the default value assigned for repetition number in the template cs_DciInfo_RandomAccessResponse is always zero.

	Summary of change
	Enhanced the template cs_DciInfo_RandomAccessResponse with parameter RepetitionNumber as a template argument and assigned the received value to RepetitionNumber inside function call fl_DciDlInfo_Init.
See also change 6.

	TTCN module
	Common\EUTRA_Templates\EUTRA_CellCfg_Templates.ttcn

	MCC160 Comment
	Accepted


Before change
	template (value) DciDlInfoCommon_Type cs_DciInfo_RandomAccessResponse(Dl_Bandwidth_Type 
                                      p_Dl_Bandwidth) := 
fl_DciDlInfo_Init(e_DlSchedulingClass_RAR, p_Dl_Bandwidth);  



After change
	template (value) DciDlInfoCommon_Type cs_DciInfo_RandomAccessResponse(Dl_Bandwidth_Type 
                                      p_Dl_Bandwidth,
                                      RepetitionNumber_Type p_RepetitionNumber := 

                                      fl_RepetitionNumber_Init()) := fl_DciDlInfo_Init(e_DlSchedulingClass_RAR, p_Dl_Bandwidth,-,p_RepetitionNumber);  



Change 6 – Correction to function ‘fl_DciDlInfo_Init’ 
	Function  name
	fl_DciDlInfo_Init ()

	Reason for change
	In this test scenario, PDSCH data needs to be repeated and this needs to be indicated in the downlink assignmnent field of “REPETITION NUMBER”. 

However, in the current TTCN implementation, there is no option to provide explicit repetition number and the default value assigned for repetition number in function fl_DciDlInfo_Init is always zero.

	Summary of change
	Enhanced the function fl_DciDlInfo_Init with parameter RepetitionNumber as a  function argument and assigned the received value to RepetitionNumber inside this function. 
Same RepetitionNumber parameter value is also passed to the initialization of template cs_DciDlInfoCommon_Init.
See also change 7.

	TTCN module
	Common\EUTRA_Templates\EUTRA_CellCfg_Templates.ttcn

	MCC160 Comment
	Accepted


Before change
	function fl_DciDlInfo_Init(DlSchedulingClass_Type p_SchedulClass,
                             Dl_Bandwidth_Type p_Dl_Bandwidth := n25,
                             PhysCellId p_PhysCellId := 0,
                             template (omit) RepetitionNumber_Type p_RepetitionNumber := 
                             fl_RepetitionNumber_Init())
    return template (value) DciDlInfoCommon_Type
  {
    //@sic R5s110034: delete sysinfoCombination parameter sic@
    //@sic R5-171566: added p_PhysCellId sic@
    var template (value) DciDlInfoCommon_Type v_DciDlInfoCommon := cs_DciDlInfoCommon_Init();
    var EUTRA_ResourceAllocationScheme_Type v_ResourceAllocationScheme := 
    fl_GetResourceAllocationScheme();  // @sic R5s110122 sic@
    var charstring v_TestcaseName := testcasename();
    var boolean v_TestcaseAttrib_3IntraFreq := f_GetTestcaseAttrib_3IntraFreq(v_TestcaseName);
    var boolean v_BLOrCE := pc_ue_CategoryDL_M1 or
                                  f_GetTestcaseAttrib_EnhancedCoverage(testcasename());

    //Legacy UE's
    if (p_SchedulClass == e_DlSchedulingClass_UeSpecific) {
      v_DciDlInfoCommon.Modulation_1stCW := qam64;
    }

    ……

    if (v_BLOrCE) {  // Overwrite
      //UE support category M1 or in enhanced coverage TC
      v_DciDlInfoCommon := cs_DciDlInfoCommon_Init;
      v_DciDlInfoCommon.RepetitionNumber := 0;
      v_DciDlInfoCommon.DCISubframeRepetitionNumber := 0;

     ……

     }
    return v_DciDlInfoCommon;
  }



After change
	function fl_DciDlInfo_Init(DlSchedulingClass_Type p_SchedulClass,
                             Dl_Bandwidth_Type p_Dl_Bandwidth := n25,
                             PhysCellId p_PhysCellId := 0,
                             template (omit) RepetitionNumber_Type p_RepetitionNumber := 
                             fl_RepetitionNumber_Init())
    return template (value) DciDlInfoCommon_Type
  {
    //@sic R5s110034: delete sysinfoCombination parameter sic@
    //@sic R5-171566: added p_PhysCellId sic@
    var template (value) DciDlInfoCommon_Type v_DciDlInfoCommon := cs_DciDlInfoCommon_Init();
    var EUTRA_ResourceAllocationScheme_Type v_ResourceAllocationScheme := 
    fl_GetResourceAllocationScheme();  // @sic R5s110122 sic@
    var charstring v_TestcaseName := testcasename();
    var boolean v_TestcaseAttrib_3IntraFreq := f_GetTestcaseAttrib_3IntraFreq(v_TestcaseName);
    var boolean v_BLOrCE := pc_ue_CategoryDL_M1 or
                                  f_GetTestcaseAttrib_EnhancedCoverage(testcasename());

    //Legacy UE's
    if (p_SchedulClass == e_DlSchedulingClass_UeSpecific) {
      v_DciDlInfoCommon.Modulation_1stCW := qam64;
    }
    ……

    if (v_BLOrCE) {  // Overwrite
      //UE support category M1 or in enhanced coverage TC
      v_DciDlInfoCommon := cs_DciDlInfoCommon_Init(p_RepetitionNumber);
      v_DciDlInfoCommon.RepetitionNumber := p_RepetitionNumber;
      v_DciDlInfoCommon.DCISubframeRepetitionNumber := 0;

     ……

     }
    return v_DciDlInfoCommon;
  }



Change 7 – Correction to template ‘cs_DciDlInfoCommon_Init’
	Template name
	cs_DciDlInfoCommon_Init ()

	Reason for change
	In this test scenario, PDSCH data needs to be repeated and this needs to be indicated in the downlink assignmnent field of  “REPETITION NUMBER”. 
However, in the current TTCN implementation, there is no option to explicitly provide the repetition number and the default value assigned for repetition number in the template cs_DciDlInfoCommon_Init is always zero.

	Summary of change
	Enhanced the template cs_DciDlInfoCommon_Init with parameter RepetitionNumber as template argument and assigned the received value to RepetitionNumber inside this template.

	TTCN module
	Common\EUTRA_Templates\EUTRA_CellCfg_Templates.ttcn

	MCC160 Comment
	Accepted


Before change
	template (value) DciDlInfoCommon_Type cs_DciDlInfoCommon_Init:=
  { 
   /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, 
   LTE_IRAT, POS) */
    Format := dci_1A,
    ResourceAllocType := ra_2_Localised,
    Modulation_1stCW := qpsk,
    Modulation_2ndCW := unused,
    FreqDomainSchedul := cs_FreqDomainSchedul(0, 0),
    RedundancyVersionList := cs_RedundancyVersionListDL_Def,
    RepetitionNumber := fl_RepetitionNumber_Init(),               // @sic R5w160205 sic@
    DCISubframeRepetitionNumber := fl_RepetitionNumber_Init()
  };



After change
	template (value) DciDlInfoCommon_Type cs_DciDlInfoCommon_Init
                                      (template (omit) RepetitionNumber_Type p_RepetitionNumber := 

                                       fl_RepetitionNumber_Init()):=
  { 
/* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */
    Format := dci_1A,
    ResourceAllocType := ra_2_Localised,
    Modulation_1stCW := qpsk,
    Modulation_2ndCW := unused,
    FreqDomainSchedul := cs_FreqDomainSchedul(0, 0),
    RedundancyVersionList := cs_RedundancyVersionListDL_Def,
    RepetitionNumber := p_RepetitionNumber,               // @sic R5w160205 sic@
    DCISubframeRepetitionNumber := fl_RepetitionNumber_Init()
  };



Change 8 – Correction to function   ‘f_GetTestcaseAttrib_DlCcchMsgInSeparateMacPdu’
	Functio name
	f_GetTestcaseAttrib_DlCcchMsgInSeparateMacPdu ()

	Reason for change
	According to TS 36.523-3, section 7.19
Due to the specific test purpose to be achieved in TC 7.1.3.5a, it is necessary to send the DL CCCH message separately.

	Summary of change
	Added TC 7.1.3.5a to the list of applicable test cases in function f_GetTestcaseAttrib_DlCcchMsgInSeparateMacPdu.

	TTCN module
	LTE_A_PRO\common\TestcaseProperties.ttcn

	MCC160 Comment
	Accepted


Before change
	function f_GetTestcaseAttrib_DlCcchMsgInSeparateMacPdu(charstring p_Testcase) return boolean
  {
    select (p_Testcase) {
      case ("TC_8_1_1_1a") { return true; }
      }
    return false;
  }


After change
	function f_GetTestcaseAttrib_DlCcchMsgInSeparateMacPdu(charstring p_Testcase) return boolean
  {
    select (p_Testcase) {
      case ("TC_8_1_1_1a") { return true; }
      case ("TC_7_1_3_5a") { return true; }
     }
    return false;
  }


5. Execution Log Files

5.1  Qualcomm MDM9207 / MDM9x07
The Qualcomm MDM 9207 / MDM 9x07 UE passed this LTE-A-PRO (Rel-13) eMTC test case on R&S CMW500 Multi-RAT Protocol Tester in LTE FDD band 4 / 13. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
Rohde-Schwarz\tc_7_1_3_5a_LTE_FDD_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

· Test Case Execution log file
Keysight\TC_7_1_3_5a_Log.html  
(Note: PICS/PIXIT settings are captured in separate file “TC_7_1_3_5a_PICS_PIXIT.xml”)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
5.2 Altair (a company of Sony Corp.) ALT 1210
The Altair (a company of Sony Corp.) ALT 1210 UE passed this LTE-A-PRO (Rel-13) eMTC test case on R&S CMW500 Multi-RAT Protocol Tester in LTE FDD band 13. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
Rohde-Schwarz\tc_7_1_3_5a_LTE_FDD_Altair.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

